
Excel y hojas de cálculo con IA

Fórmulas que funcionan a la primera porque sabes pedirlas, y resultados
que sabes comprobar
Edición: 05/11/2026. Para quién: trabajas en administración, por tu cuenta o al frente de un equipo, y pasas
parte del día en Excel o en Hojas de cálculo de Google. Sabes hacer una suma, filtrar y copiar una fórmula hacia
abajo; no programas. Quieres que la IA te escriba fórmulas, te ayude a limpiar exportaciones y a resumir tablas,
sin que se cuelen errores que no ves.

Alcance
Los nombres y la sintaxis de las funciones son los de Excel en español, comprobados en las páginas de soporte
de Microsoft el 05/11/2026, con el punto y coma como separador. Lo que se dice de Copilot en Excel, de
Gemini en Hojas de cálculo y de los planes que los incluyen es lo que publicaban ese día las páginas oficiales
de Microsoft y Google; estas funciones cambian a menudo, así que cada capítulo enlaza su fuente para que la
compruebes. Las cifras del laboratorio son de pruebas propias del 04/11/2026, con el método publicado en
beautythings.net/lab. Es el único aviso de este tipo que vas a leer.

Cómo usar este libro
Haz el capítulo 01 entero, con los dos archivos del paquete: en menos de una hora tendrás una fórmula hecha
por IA funcionando y el cuadre que te avisa cuando un resultado está mal aunque no haya ningún error. Todo
lo demás se apoya en eso. Después, ve a lo que necesites:

Lo que quieres Capítulo

Que la IA deje de darte fórmulas en inglés, con comas o con
funciones que tu Excel no tiene

02

Limpiar una exportación: espacios, números como texto, fechas
raras, duplicados

03

Un resumen por zona y mes, una tabla dinámica, un gráfico y su
comentario

04

Subir la hoja a un chat sin sacar datos que no debes 05

Saber qué hacen hoy Copilot en Excel y Gemini en Hojas de cálculo,
y con qué plan

06

Revisar un informe antes de entregarlo 07

Automatizar una tarea repetida con una macro o con Apps Script 08

Qué necesitas
• Excel para Microsoft 365, 2021 o 2024. Con 2016 o 2019 funciona casi todo; el capítulo 02 dice qué

funciones no tendrás y qué usar en su lugar. Hojas de cálculo de Google sirve para los capítulos 02 a 08.

• Cualquier chat de IA, gratuito o de pago.

• El paquete: ventas-ejemplo.csv y clientes-ejemplo.csv (datos ficticios), la ficha de la hoja, las cifras
de control, 25 prompts y las dos macros de ejemplo.

01 · La primera hora: una fórmula que
funciona a la primera y un cuadre que te avisa
Vas a pedirle a una IA una fórmula, la vas a pegar y va a funcionar a la primera. Después vas a descubrir que el
resultado está mal aunque la fórmula esté bien, y vas a montar en dos minutos la comprobación que lo caza.
Esa comprobación es lo más útil de todo el libro: la usarás cada vez que una IA (o un compañero) te entregue
números.

Necesitas Excel (Microsoft 365, 2021 o 2024; en el capítulo 02 tienes qué hacer si tu versión es más antigua) y
los dos archivos del paquete: ventas-ejemplo.csv y clientes-ejemplo.csv. Son datos ficticios: 49 líneas
de ventas de un trimestre y 8 clientes. Cualquier chat de IA sirve, gratuito o de pago. No vas a subir ningún
archivo; solo vas a describir la hoja.

Paso 1 (10 minutos): prepara el libro
1. Abre ventas-ejemplo.csv con doble clic. Si cada fila aparece entera en la columna A, con punto y coma,

ciérralo sin guardar y ábrelo desde Datos > Obtener datos > Desde archivo > Desde texto/CSV, con
delimitador «Punto y coma».

2. Abre clientes-ejemplo.csv igual.

3. En el libro de clientes, clic derecho en la pestaña de la hoja > Mover o copiar > «Al libro»: el de ventas >
marca Crear una copia > Aceptar.

4. Cambia el nombre de las pestañas a Ventas y Clientes. Guarda como libro de Excel (.xlsx), por
ejemplo trimestre.xlsx.

Comprueba: en Ventas hay datos de la fila 2 a la 50, columnas A a G (Fecha, Factura, Codigo_cliente, Producto,
Unidades, Precio_unitario, Importe). En Clientes, de la fila 2 a la 9, columnas A a D (Codigo_cliente, Cliente,
Zona, Comercial).

Paso 2 (10 minutos): la ficha de la hoja
Un chat no ve tu Excel. Si le pides «una fórmula para traer la zona de cada cliente», se la inventa con columnas
imaginarias, nombres de funciones en inglés y comas entre argumentos. Pegas eso en un Excel en español y te
devuelve un error, o algo peor: una fórmula que funciona sobre las celdas equivocadas.

Hay cuatro datos que la IA no puede adivinar y que cambian la fórmula:

Dato Por qué importa En este ejemplo

Idioma de Excel Las funciones se llaman distinto: XLOOKUP es
BUSCARX

Español

Separador de argumentos En Excel en español, la documentación
oficial de Microsoft escribe las fórmulas con
punto y coma

;

Versión BUSCARX no existe en Excel 2016 ni 2019 Microsoft 365 (o la tuya)

Estructura exacta Hoja, columnas, primera y última fila, qué es
texto y qué es número

La de abajo

Para saber tu separador, escribe en una celda =SUMA( y mira la ayuda que aparece debajo: si dice
SUMA(número1; [número2]; ...), es punto y coma.

Con esos datos, escribe la ficha. La plantilla está en plantillas\ficha-de-la-hoja.md; rellenada queda así:

Trabajo con Excel para Microsoft 365 en español. Separador de argumentos: punto y coma.
Hoja "Ventas", filas 2 a 50, fila 1 = encabezados:
  A Fecha (fecha), B Factura (texto, p. ej. F-2026-0001), C Codigo_cliente (texto, p. ej. C001),
  D Producto (texto), E Unidades (número), F Precio_unitario (número), G Importe (número, euros).
Hoja "Clientes", filas 2 a 9, fila 1 = encabezados:
  A Codigo_cliente (texto), B Cliente (texto), C Zona (Norte, Sur, Centro o Este), D Comercial (texto).

Paso 3 (10 minutos): pide la fórmula
Debajo de la ficha, en el mismo mensaje:

Quiero en Ventas!H2 la zona del cliente de esa fila, buscándola por Codigo_cliente en Clientes.
La copiaré hacia abajo hasta H50, así que fija los rangos de Clientes.
Si el código no existe en Clientes, que escriba "SIN ZONA" en lugar de un error.
Dame solo la fórmula, con nombres de funciones en español y punto y coma, y una frase que explique cada 
argumento.

La respuesta correcta es esta (o equivalente):

=BUSCARX(C2;Clientes!$A$2:$A$9;Clientes!$C$2:$C$9;"SIN ZONA")

Compárala con la sintaxis oficial de Microsoft: BUSCARX(valor_buscado; matriz_buscada;
matriz_devuelta; [si_no_se_encuentra]; [modo_de_coincidencia]; [modo_de_búsqueda]). Busca
C2 en la columna de códigos de Clientes y devuelve lo que haya en la misma fila de la columna de zonas. El
cuarto argumento, «SIN ZONA», es lo que va a hacer visible el problema dentro de un momento. El modo de
coincidencia por defecto es la coincidencia exacta, que es lo que quieres con códigos.

Escribe Zona en H1, pega la fórmula en H2 y haz doble clic en el cuadradito de la esquina inferior derecha de la
celda para copiarla hasta H50.

Si la IA te dio XLOOKUP, comas o rangos sin $, no la arregles a mano: dile qué ha fallado y que la repita. Así
aprendes qué tienes que poner en la ficha la próxima vez.

Paso 4 (10 minutos): el resumen por zona
Crea una hoja Resumen. En A1 escribe Zona y en B1 Importe. En A2:A5, Norte, Sur, Centro y Este. Pide al
chat:

En la hoja Resumen, A2:A5 tiene las zonas. Quiero en B2 la suma de Ventas!G2:G50 para las filas
cuya zona (Ventas!H2:H50) sea la de A2. La copiaré hasta B5.

Respuesta esperada:

=SUMAR.SI.CONJUNTO(Ventas!$G$2:$G$50;Ventas!$H$2:$H$50;A2)

Fíjate en el orden: en SUMAR.SI.CONJUNTO el rango que se suma va primero; en SUMAR.SI, va el tercero.
Microsoft lo señala como causa frecuente de errores, y las IA lo confunden cuando mezclan las dos funciones.

Copia la fórmula hasta B5. Tienes cuatro cifras plausibles. Un informe normal acabaría aquí.

Paso 5 (10 minutos): el cuadre
Añade tres filas en Resumen:

Celda Escribe Qué es

A7 / B7 Suma de zonas / =SUMA(B2:B5) Lo que reparte tu resumen

A8 / B8 Total de Ventas /
=SUMA(Ventas!G2:G50)

Lo que hay en origen

A9 / B9 Diferencia / =B8-B7 Tiene que ser 0

La diferencia no es 0: son 279,30 €. El resumen se ha dejado dinero fuera y ninguna celda muestra un error. Las
dos fórmulas son correctas. Lo que está mal son los datos.

Para ver qué filas faltan, en cualquier celda libre de Resumen:

=FILTRAR(Ventas!B2:H50;Ventas!H2:H50="SIN ZONA")

Salen dos facturas: F-2026-0010, de 73,80 €, y F-2026-0023, de 205,50 €. Sus códigos parecen C002 y C004, que
sí existen en Clientes. No coinciden porque llevan algo invisible detrás.

Paso 6 (10 minutos): el arreglo y la segunda sorpresa
Pide al chat que limpie el código antes de buscarlo. Lo habitual es que te proponga ESPACIOS:

=BUSCARX(ESPACIOS(C2);Clientes!$A$2:$A$9;Clientes!$C$2:$C$9;"SIN ZONA")

Cópiala hasta H50. La diferencia baja a 205,50 €: F-2026-0010 ya encuentra su zona, F-2026-0023 sigue sin ella.
La documentación de Microsoft explica por qué: ESPACIOS quita el espacio normal (carácter 32), pero no el
espacio de no separación (carácter 160), que es muy común en datos copiados de una web, de un PDF o de
otro programa. Por sí sola, ESPACIOS no lo quita.

La fórmula que funciona sustituye primero el carácter 160 por un espacio normal y luego recorta:

=BUSCARX(ESPACIOS(SUSTITUIR(C2;CARACTER(160);" "));Clientes!$A$2:$A$9;Clientes!$C$2:$C$9;"SIN ZONA")

Cópiala hasta H50. Ahora la diferencia es 0 y el resumen queda así:

Zona Importe

Norte 2.390,15 €

Sur 5.537,20 €

Centro 4.494,65 €

Este 4.141,65 €

Suma 16.563,65 €

Si tus cifras coinciden, ya tienes el método completo: ficha, petición precisa, fórmula, cuadre.

Lo que acabas de aprender
Una IA puede escribir una fórmula perfecta y el resultado salir mal, porque el problema estaba en los datos y
nadie se lo dijo. En nuestro laboratorio pasó lo mismo con documentos: con la misma acta y los mismos cinco
modelos, un encargo ingenuo superó 85 de 120 criterios y un encargo con reglas explícitas, 120 de 120
(beautythings.net/lab/encargo-pesa-mas-que-el-modelo). La mejora vino de cómo se pidió el trabajo, no del
modelo. La ficha de la hoja es ese encargo con reglas, aplicado a Excel.

El cuadre es tu seguro: un total que se puede comparar con otro total calculado por otro camino. Cuesta tres
celdas y caza los errores que no dan error.

Quedan más trampas en este archivo: una factura duplicada, unas unidades escritas como texto, una fecha que
Excel ha leído mal sin avisar y un importe que no cuadra con su precio. Las encontrarás en los capítulos 03 y 07.
Las cifras de control de todas ellas están en plantillas\comprobaciones.md.

Fuentes
• Microsoft, «Función BUSCARX» (sintaxis, coincidencia exacta por defecto, no disponible en Excel 2016 ni

2019): https://support.microsoft.com/es-es/Excel/functions/xlookup-function

• Microsoft, «Función SUMAR.SI.CONJUNTO» (orden de argumentos frente a SUMAR.SI):
https://support.microsoft.com/es-es/office/sumar-si-conjunto-funci%C3%B3n-sumar-si-conjunto-c9e748f5-7ea7-455d-9406-611cebce642b

• Microsoft, «Función ESPACIOS» (no quita el espacio de no separación, valor 160):
https://support.microsoft.com/es-es/office/espacios-funci%C3%B3n-espacios-410388fa-c5df-49c6-b16c-9e5630b479f9

• Microsoft, «Función FILTRAR»: https://support.microsoft.com/es-ES/Excel/filter-function

• Laboratorio BeautyThings, «¿Importa más el modelo o cómo le pides el trabajo?», 04/11/2026:
https://beautythings.net/lab/encargo-pesa-mas-que-el-modelo


